Dynamic Processing of Analog Data Using an Apple Computer[1]

William H. Caskey

Soil Microbiologist, ARS, USDA
Morris, MN 56267

Abstract

GASCHROMATOGRAPH is a program which supports data acquisition from laboratory instruments using an Apple-ADALAB system with dynamic calculations of areas under the peaks for quantitative analysis. Requirements of the system are: an Apple II computer with 48 K RAM, monitor, disk drive, printer, and game controller. Analog to digital conversion of the instrument signal with 13 bits of resolution is accomplished with an ADALAB interface board. The areas calculated by GASCHROMATOGRAPH are shown to be essentially identical to those determined by direct digitization of the area under a peak produced by the tracing on a strip chart recorder. The integration algorithms of GASCHROMATOGRAPH were designed for and are applicable to analysis of samples which are resolved completely by the chromatography system.

Introduction

Quantitative analysis using instruments producing an analog signal requires calculation of peak height or area under a peak from the digitized signal. Microcomputers equipped with a high resolution analog to digital (A / D) converter would be ideally suited for such use, but A / D converters with sufficient resolution and appropriate software for processing the digitized data have not been available. The purpose of this paper is to describe software that permits an Apple II[2] computer equipped with a commercially available A / D converter to be used to collect and process data from a gas chromatograph for the analysis of nitrogen, nitrous oxide (N_2O), and carbon dioxide. The peak-picking algorithm is applicable only to analysis of chromatograms with well-separated peaks.

Hardware

The microcomputer system consists of an Apple II Plus computer with 48 K RAM, monitor, one disk drive, and a printer. Pushbutton #1 of a joystick game controller is used as a remote start switch. The ADALAB[3] interface board contains an Intersil 7109 dual slope integrating A / D converter operating with 12 or 13 bits of resolution (3). A resistor capacitor oscillator on the board sets the A / D conversion rate nominally at 20 conversions per second. It has a precision of 0.025% and overall accuracy of 0.1%. The interface board has full scale voltage inputs of 0.5, 1.0, 2.0, and 4.0 V which are jumper selectable. The board also contains two MOS Technology 6522 Versatile Interface Adapters (VIA) each of which contains two 16-bit timers.

The signal from the gas chromatograph has a nominal 1V maximum. The actual maximum voltage was measured as 1.33 V. A resistor bias circuit was installed on the signal cable to adjust the maximum voltage to 0.5005 V. With the 0.5 V maximum input selected on the ADALAB interface card, detector saturation can be detected with the overflow status of the A / D converter while using >99% of the full deflection range of the detector.

*Received April 1983.

[1]Contribution from the North Central Soil Conservation Research Laboratory, Agricultural Research Service, USDA, Morris, MN, in cooperation with the Minnesota Agric. Exp. Sta., Sci. Series 13,444.

[2]Apple is a registered trademark of Apple Computer Inc., Cupertino, CA.

[3]ADALAB is a registered trademark of Interactive Microware Inc., State College, PA. Products are mentioned by name for the convenience of the reader and their mention does not constitute an endorsement by USDA.

Software

The program GASCHROMATOGRAPH (Listing 1) is written in Forth-79 implemented for the Apple II by MicroMotion (West Los Angeles, CA). This Forth conforms to the International Forth-79 Standard (2), but contains additional words to support functions unique to and optimized for the Apple II computer. Extensions for double numbers, strings, and high resolution graphics are required to compile the program (1, 2). Compilation takes about 1 minute. MicroMotion Forth supplies a word TURNKEY which allows creation of an application program which begins execution when the dictionary is loaded.

GASCHROMATOGRAPH is menu driven with the prompt line appearing at the bottom of a blank screen.

1=SMPL 2=STND 3=BSLN 4=VRBL 5=LBLS

Respectively, these selections allow the user to analyze a sample, analyze a standard, adjust the base line, change the integration parameters, or to date and name an experiment. Upon completion of any of these functions, control is returned to this prompt line. Selecting SMPL, STND, or BSLN initiates the data acquisition functions (SCR #48). A prompt is displayed and the computer waits until Pushbutton #1 of the game controller is pressed before data acquisition actually begins. A sample analysis is ended by pressing any key on the keyboard. Results are printed with sample numbers automatically incremented. Provision is made for the sample number to be changed before printing in case a sample needs to be reanalyzed or to initiate sample number incrementing from a value other than 1. Selection of SMPL, STND, or LBLS also invokes the appropriate printing function. The printer commands in SCR #47 are specific for a $Base_2$ printer ($Base_2$, Inc., Fullerton, CA) with a parallel interface (Apple Computer Inc., Cupertino, CA) and will not necessarily drive another printer.

An example of the report produced by the printing functions is shown in Figure 1. For each sample or standard injected, the report includes the peak number (peaks are successively numbered as the commands elute), the retention time of the peaks for identification, the width of the peak from shoulder to shoulder, and the area of the peak.

PHASE I AUG. WITH C2H2 .2ML
9/2/82

SAMPLE #1

PK #	PET. TIME	PK WIDTH	APEA
1	380	99	40934
2	325	79	1920

SAMPLE #2

PK #	PET TIME	PK WIDTH	AREA
1	403	154	99400*
2	680	29	369
3	828	104	3480

SAMPLE #3

PK #	PET TIME	PK WIDTH	AREA
1	400	149	80148
2	825	89	2568

STD=1&2 1.25ML

PK #	PET TIME	PK WIDTH	AREA
1	360	39	1719
2	645	49	2488
3	825	89	2579
4	1110	89	2619

Figure 1. Sample printing of the results produced by GASCHROMATOGRAPH. Experiment and sample numbers are emphasized. Retention time and peak width are in units of 0.1 sec. An asterisk after an area indicates detector overflow.

Functions for initiation of the timer and A/D converter were written in machine language (SCR #41-42) using the resident assembler in the Forth-79 implementation.[4] SETCLOCK connects the two timers on one 6522 VIA for use as a 32 bit countdown timer setting the A/D sampling frequency at 10 Hz. ADTIM initializes the A/D converter to operate in slot #2 at 12 bits resolution and the thirteenth bit as an overflow indicator, giving a resolution of 1:4096 for full scale voltage deflection. The digitized value of the signal is processed through the variable DATUM.

As the analog signal is sampled at 100 msec intervals, each A/D conversion is plotted on the CRT display from left to right, eliminating the need for a strip chart recorder. The high resolution graphics display on the Apple II is 280 points horizontally by 192 points vertically. When the chromatogram display reaches the right edge of the screen, it begins again at the left erasing the previous tracing. Therefore, at the sampling intervals of 100 msec, the data from the last 28 seconds are visible on the screen. If the digital value of a discrete analog sample is greater than 191, it is not plotted. Thus, the display will show a peak going up and returning to the baseline, but may not show the top of the peak. A reference line is drawn at Y = 0 when the digitized signal is greater than 0 but is not drawn when it goes below 0. The line thus serves as an indicator of the position of the tracing when there is no point plotted on the screen.

Two parameters control the integration algorithm: data bunching and slope sensitivity. Data bunching is the number of individual A/D readings that are summed and processed as a datum by the integration algorithm. The value for slope sensitivity is the deviation from the baseline that must occur for the beginning of a peak to be recognized. Also, a peak must return to a value within slope sensitivity of the baseline (at the beginning of the peak) before the end of the peak is recognized. A third parameter, peak width, is the minimum width of a peak (in units of 0.1 sec) that will be included on a final report of the integrated chromatogram. Values for slope sensitivity and peak width may vary from 1 to 99, or 0.025% to 2.475% of full scale for slope sensitivity and 0.1 to 9.9 sec for peak width. For the gas analyses described, parameters are 5, 10, and 10 for data bunching, slope sensitivity, and peak width, respectively. These parameters can be changed by selecting the VRBL option on the menu. The last used set of parameters are stored on diskette (SCR #60). The peak number, retention time in units of 0.1 sec, peak width in units of 0.1 sec, and the area under the peak are printed after the completion of data acquisition (Fig. 1). If the amount of a compound injected resulted in a peak too large for the detector causing the detector to overflow, an asterisk is printed next to the area (Fig. 1). If no peaks are detected, a message of "NO PEAKS" is printed. The arrays allotted in SCR #44 restrict to 11 the number of peaks allowable in a single chromatogram. Also, the size of the array for storing the "data bunches" during data acquisition restricts the time (in minutes) for a single chromatographic analysis to 5.120 divided by 600 multiplied by DATABUNCH value. Since DATABUNCH can vary from 1 to 7, maximum time for data acquisition varies from 8.5 to 59.7 minutes.

The algorithms for recognition of a peak and calculation of the area under a peak are in SCR #44-46 and are illustrated in Fig. 2. The baseline (BSLN) is calculated as the average of DATABUNCH individual A/D conversions and is updated for each "data bunch" as long as no peak has been detected. The primary datum used for the integration routines is ARINC (for area increment) which is the sum of DATABUNCH successive A/D conversions minus BSLN. If ARINC is $\geq$ BSLN + SS (slope sensitivity), a flag is set indicating a peak has been detected and summation of the area under the peak begins. Because the A/D converter on the ADALAB interface card is an integrating type, the area can be calculated by summing the successive ARINC's. Although calculated using ARINC, peak area is effectively calculated by:

[4]These machine language routines can be easily formulated from examples in the documentation accompanying the ADALAB card. Permission to publish these routines was granted by Interactive Microwave, Inc., State College, PA.

$$\text{PKAREA} = \sum_{t_B}^{t_E} (\text{DATUM} - \text{BSLN})$$

where DATUM = the value of a single A D conversion,

BSLN = the baseline as calculated immediately prior to detection of the peak.

t_B = elapsed time of A D conversion when a peak is recognized (DATUM $\geq$ BSLN + SS), and

t_E = elapsed time of A D conversion which signals the end of the peak (DATUM $\leq$ BSLN + SS).

For the example in Fig. 2, the integrated area is represented by the shaded area. For the area of a peak to be included in the final report, the difference between t_E and t_B must be greater than PEAKWIDTH. Since the maximum value for DATABUNCH is 7 and the maximum A D conversion value is 4096, ARINC and all other integration variables are single precision variables. The successive ARINC's are summed into the double precision variable PKAREA and stored in the double precision array PEAKAREA for the printed report. Using single precision arithmetic for all calculations except PKAREA and PEAKAREA allows exploitation of the execution speed of Forth to provide dynamic calculation of areas while the data are being collected.



Figure 2. Hypothetical peak illustrating the area calculated by the integration algorithms and the parameters that control the integration.

Results

The accuracy of the algorithm used to calculate the area under a peak was verified by analyzing a range of amounts of N_2O of 5 to 25 nanoliters. The instrument used for analysis was a Varian 3700 gas chromatograph equipped with a ^{63}Ni electron capture detector operated at 320°C. Separation was achieved with a 2m x 3mm Poropak Q column at 75°C. Argon-methane (95%-5%) was the carrier gas at a flow of 40 ml min^{-1}. The signal produced by the peaks was simultaneously measured with the Apple-

ADALAB system and by tracing the analog output on a strip chart recorder. The analog tracing was then digitized and the area in mm^2 was determined using a Hewlett Packard 1000 minicomputer with a Hewlett Packard 9874A digitizer. Also, known voltages were measured by both techniques to calibrate the area units for each method. The results of that comparison are shown in Table I. The analog signal strength was calculated from the calibration value and the response in mV•min vs amount of N_2O was analyzed by linear regression. The slopes for the Apple-ADALAB system and the recorder were 0.601 and 0.605, respectively. The intercepts were 1.188 and 1.165, respectively. Comparison of the slopes and intercepts using the t test showed that neither was significantly different (99% confidence level).

Discussion

GASCHROMATOGRAPH can be used on an Apple-ADALAB system connected to any instrument producing an analog signal which utilizes peak areas as the basis for quantitative analysis and when the peaks are well separated. The areas calculated by the integration algorithms in the program produced results that were comparable to those obtained by direct measurement of the area under a peak traced on strip chart recorder. The area calculated using the digitizer generally was slightly greater than that reported by the Apple-ADALAB system because the integration parameters of slope sensitivity and data bunching allowed less of the beginning and ending shoulders of the peaks to be included than did direct digitization of the recorder tracing. The differences were not significant. The equality of the results indicated that aliasing of the signal was inconsequential.

One problem with GASCHROMATOGRAPH prohibiting its use in general chromatography applications is the inability of the integration scheme to resolve fused peaks. For the end of a peak to be recognized, the signal must drop to a value that is within the slope sensitivity parameter of the original baseline. A similar problem would result if positive baseline drift occurred while a compound was eluting, resulting in a new baseline at a level greater than the slope sensitivity value above the original baseline. However, GASCHROMATOGRAPH adjusts for baseline drift when peaks are not being detected. Occurrence of such problems will be reported as excessively large values for peak width on the final report. Resolution can be increased by varying the parameters of slope sensitivity and data bunching. The inability to resolve fused peaks is not a problem in the gas analyses for which GASCHROMATOGRAPH was written.

Although the A/D converter on the ADALAB interface will convert analog signals over a range of ± the selected full scale voltage, GASCHROMATOGRAPH supports only the positive portion of this range. Negative voltages are not plotted on the CRT and are not integrated.

Other revisions to GASCHROMATOGRAPH could make the program more generally applicable. With the MAX function of Forth-79, peak heights could be determined and included in the final report. Since the raw data are currently stored in the array DATA at the end of a sample analysis, inclusion of a provision for reintegration using different integration parameters would be simple. Also, the raw data could be stored on diskette and recalculated at a later time. Furthermore, with the chromatogram stored in DATA or on a diskette, an algorithm could compress the chromatogram to fit the high resolution display and a permanent record of the chromatogram made using a printer capable of reproducing an image of the high resolution display. If the dynamic integration capability is the primary factor in using this program, storage of the raw data in the array DATA could be eliminated since the current integration algorithm requires only the previous ARINC and the current ARINC. This change would decrease the amount of memory required for the application, but more importantly would create an infinite run time for an individual sample. This would be particularly important in the use of this system for gas chromatography of organic compounds requiring long analytical times. If the number of expected peaks is greater than 11, the size of the array in which the integrated results are stored can be increased (SCR #44).

GASCHROMATOGRAPH permits the use of the Apple-ADALAB system to collect analog data from analytical instruments and provides dynamic calculations of areas under the peaks for quantitative analysis. This system provides the capability of computing integrators but with the added flexibility for further analysis inherent in a small computer system. The potential exists for the sample to be converted into units of concentration or amounts by comparing their integrated areas with the areas of peaks produced from injected standards and storing that data on diskette for recall by other programs used for more complete data analysis. This would eliminate the inefficiency of manual translation of data and the errors that frequently occur during such translation.

Acknowledgement

The author acknowledges the technical assistance of Steve Van Kempen and Steve Masterson and the assistance of Owen Devine for statistical analysis.

References

1. Anderson, A., S. Tabor, M. Tracy, and P. Wasson. 1982. *Forth-79 Floating Point Arithmetic and HIRES Turtlegraphics Manual, Apple II Version.* MicroMotion, West Los Angeles, CA, 20 p.
2. Anderson, A., and M. Tracy. 1981. *Forth-79 Tutorial and Reference Manual, Apple II Version, Volume I.* MicroMotion, West Los Angeles, CA, 198 p.
3. Bogdan, M. R., and P. K. Warme. 1983. "Personal Computers in the Laboratory. Part I. Hardware." *Computer Applications in the Laboratory* 1:40-44.

Dr. Caskey graduated with a BS from Louisiana Technical University in 1967, an MS from Texas A and M in 1975 and a PhD from Michigan State University in 1978, all in microbiology. He is interested in microbial nitrification and denitrification of soils. Currently he is studying the effects of reduced tillage farming on rates of nitrification and denitrification, and their impact on fertilizer practices.

TABLE I

Comparison of Integrated Area From the Apple-ADALAB System Using GASCHROMATOGRAPH With Digitized Area From the Tracing of a Strip Chart Recorder

Amount of N_2O (nl)	**Area**		**Analog signal (mV•min)***	
	Apple (units)	**Recorder** (mm^2)	**Apple**	**Recorder**
5	11800 ± 207†	393 ± 6	4.2	4.2
10	19428 ± 64	645 ± 9	6.9	7.0
15	29125 ± 91	968 ± 17	10.4	10.5
20	36927 ± 153	1235 ± 14	13.2	13.3
25	45222 ± 139	1499 ± 30	16.2	16.2

*Calculated using calibration factors of 3.57×10^{-4} mV•min $unit^{-1}$ for the Apple and 1.08×10^{-2} mV•min mm^{-2} for the recorder.

†Average ± standard deviation of three measurements.

TABLE 2

Glossary of Nonstandard Forth Words Supplied by MicroMotion Forth-79 Used in GASCHROMATOGRAPH (1, 2)

ARRAY	n ---
Creates a dictionary entry and allots n two-byte elements.	
Used as n ARRAY < name >.	
BLACK	n ---
Sets color of display to black (on black).	
CARRAY	n ---
Creates a dictionary entry and allots n one-byte elements.	
Used as n CARRY < name >.	
CH	n ---
Positions the cursor to column n.	
CV	n ---
Positions the cursor to row n.	
FULL	n ---
Selects full-screen high resolution graphics.	
GODO	n ---
Used as GODO word0 wordl .... wordx THEN. The nth word between GODO and THEN will be executed.	
HGR	
Selects high resolutions graphics mode with four lines of text at bottom of display.	
HOME	---
Clears the display and positions cursor in upper left-hand corner.	
HPLOT	---
Plots a point at coordinates x, y.	
PR#	x y ---
Select interface in slot n as output device.	
STRING	n ---
Creates a string variable with a maximum length of n characters.	
Used as n STRING < name >.	n ---
TX	
Returns from high resolution graphics to normal text mode.	
WHITE	---
Sets color of display to white (on black).	

Listing 1. Forth code for GASCHROMATOGRAPH. SCR#41-42 apply only to the ADALAB interface card and SCR#47 contains printer-specific commands (see text). The listing includes several extensions to Forth that are included in MicroMotion Forth-79 for the Apple II (see Table 2).

```
SCR # 41
( ADALAB FUNCTIONS)

HEX
VARIABLE DATUM
CODE SETCLOCK                        ( TIME -- )
        E0 # LDA, C20B STA, BE # LDA, C204 STA,
        C7 # LDA, C205 STA, BOT LDA, C208 STA,
        BOT 1+ LDA, C209 STA, POP JMP, END-CODE
CODE ADTIM                           ( TIME-DELAY -- )
        1 L: C208 LDA, BOT CMP, 1 L# BNE,
             C209 LDA, BOT 1+ CMP, 1 L# BNE,
             8A # LDA, C20C STA, C200  STA,
        2 L: C20D LDA, 10 # AND, 2 L# BEQ,
             C210 LDA, DATUM STA, C220  LDA,
             DATUM 1+ STA, 20 # AND, 4 L# BNE,
 -->

SCR # 42
( ADALAB FUNCTIONS CONTINUED)

        3 L: DATUM LDA, FF # EOR, DATUM STA,
             DATUM 1+ LDA, FF # EOR, E0 # ORA,
             DATUM 1+ STA, POP JMP,
        4 L: DATUM 1+ LDA, 1F # AND, DATUM 1+
             STA, POP JMP, END-CODE
DECIMAL

 -->

SCR # 43
( STRIP CHART RECORDER SIMULATOR)

280 CARRAY GRAPH
: INITGRAPH 0 GRAPH 280 0 FILL ;
: GRAPHIT                            ( 1 x DATUM -- 1 DATUM)
        SWAP 2DUP DUP GRAPH C@ BLACK HPLOT
        0< IF SWAPDROP 0 2DUP HPLOT SWAP ELSE DUP 0 WHITE
           HPLOT THEN OVER DUP 191 <
        IF 2DUP SWAP GRAPH C! HPLOT ELSE 2DROP THEN ;
 -->
```

```
SCR # 44
( DATA STORAGE AND INTEGRATION VARIABLES & COMMANDS )

VARIABLE DATABUNCH VARIABLE PW VARIABLE SS
VARIABLE ARINC 0 ARINC ! VARIABLE BSLN
5120 ARRAY DATA
: INITDATA 0 DATA 10240 0 FILL ;
: 2ARRAY CREATE 4 * ALLOT DOES> SWAP 4 * + ;
11 ARRAY PKWID 11 ARRAY MXM 11 ARRAY OVFL 11 2ARRAY PEAKAREA
VARIABLE PKST? 0 PKST? ! VARIABLE PK# 2VARIABLE PKAREA
VARIABLE STRT VARIABLE END VARIABLE OV 32 OV !
8190 CONSTANT TOOBIG 4096 CONSTANT DATAMAX
: INITINTEGRATE 0 PK# ! 0. PKAREA 2! 0 STRT ! ;
: SLICE# DATABUNCH @   ;
-->


SCR # 45
( INTEGRATION ROUTINES )

: SUMPEAK S->D PKAREA 2@ D+ PKAREA 2! ;
: PEAKDONE                              ( I DATA--I )
       END @ DUP STRT @ - DUP PW @ >
       IF DUP PK# @ 1+ DUP PK# ! DUP DUP PKAREA 2@ ROT
            PEAKAREA 2! OV @ SWAP OVFL ! 2DUP PKWID ! SWAP 2
            STRT @ + SWAP MXM !
       THEN DROP STRT ! 32 OV ! 0 PKST? ! 0. PKAREA 2! DROP ;
: PEAKEND?                              ( I DATA--I )
       DUP DUP SS @ < SWAP 4 PICK SLICE# 1- DATA @ =
       OR IF PEAKDONE ELSE SUMPEAK DUP END ! THEN ;
: PEAKMAX?                              ( I DATA--I )
       1 PKST? ! 2DUP SWAP SLICE# 1- DATA @ SWAP >
       IF PEAKEND? ELSE SUMPEAK THEN ;
-->


SCR # 46
( INTEGRATION ROUTINES CONTINUED)

: PEAKSTART?                              ( I DATUM-BSLN--I )
       ARINC @ + 2DUP SWAP SLICE# DATA ! 0 ARINC ! DUP
       SS @ > PKST? @ OR IF PEAKMAX? ELSE BSLN @ DATABUNCH @
                                    DUP ROT * ROT + / BSLN ! DUP 1+
                                    STRT ! THEN ;
: INTEGRATE                               ( I DATUM--I)
       OVER 1 = IF DUP BSLN ! THEN
       DUP TOOBIG > IF 42 OV ! DROP DATAMAX THEN BSLN @ -
       OVER DATABUNCH @ MOD IF ARINC +! ELSE PEAKSTART? THEN ;
-->
```

```
SCR # 47
( PRINTER COMMANDS)

: SCRRESET HOME 22 CV ;
: PRINTERON HOME 1 PR# 9 EMIT . " 80N" 58 27 EMIT EMIT 10 57 27
        EMIT EMIT EMIT CR ;
: PRINTEROFF 9 EMIT ." 40N" 0 PR# HOME ;
: PRINTHEADING CR 20 SPACES 14 EMIT TYPE CR CR 18 SPACES
        ." PK#" 4 SPACES ." RET. TIME" 3 SPACES ." PK WIDTH"
        5 SPACES  ." AREA" CR CR ;
: PRINTRESULTS PRINTERON PRINTHEADING PK# @
        DUP 0= IF 25 SPACES ." NO PEAKS" CR CR DROP EXIT THEN
        1+ 1 DO I DUP 18 SPACES 3 .R DUP MXM @ 10 .R
                DUP PKWID @ 11 .R DUP PEAKAREA 2@ 13 D.R OVFL @
                EMIT CR
            LOOP CR CR ;
 -->


SCR # 48
( DATA ACQUISITION COMMANDS)

VARIABLE TIME 32767 TIME !
: KBDRESET 0 -16368 C! ;
: GR&TIMMODE TIME @ SETCLOCK HGR FULL ;
: TAKEDATA 1+ DUP 280 MOD TIME @ 1- DUP TIME ! ADTIM DATUM @ ;
: START HOME INITGRAPH INITDATA INITINTEGRATE
        12 CV 6 CH ." PRESS START BUTTON TO START"
        13 CV 10 CH ." AND ANY KEY TO STOP"
        BEGIN 1 PBUTTON UNTIL CR ;
 -->


SCR # 49
( INTEGRATION PARAMETER COMMANDS)

: TOASCII 10 /MOD 48 + SWAP 48 + SWAP ;
: ACCEPT 60 BLOCK DROP UPDATE 1 ;
: PRINTCHNG CR 22 CV ." ENTER NEW VALUE FOR " ;
: CHNGSS PRINTCHNG ." SLOPE SENSITIVITY " INPUT DROP DUP
        DATABUNCH @ * SS ! TOASCII 60 BLOCK 128 + C! 60 BLOCK
        129 + C! 0 ;
: CHNGPW PRINTCHNG ." PEAK WIDTH " INPUT DROP DUP PW ! TOASCII
        60 BLOCK 131 + C! 60 BLOCK 132 + C! 0 ;
: CHNGDB PRINTCHNG ." DATA BUNCHING " INPUT DROP DUP SS @
        DATABUNCH @ * SS ! DUP  DATABUNCH ! TOASCII
        60 BLOCK 134 + C! 60 BLOCK 135 + C! 0 ;
 -->
```

```
SCR # 50
( INTEGRATION PARAMETER MENU)

: CHOICE 15 CV ." TO ACCEPT DEFAULT VALUES, PRESS  RETURN" CR CR
              ." TO CHANGE, ENTER LETTER OF FUNCTION   "
      KEY 64 - GODO ACCEPT CHNGDB CHNGSS CHNGPW THEN ;
: PARAMETERMENU BEGIN HOME 10 CV
      6 CH ." (A) DATA BUNCHING      = " DATABUNCH @ DUP . CR
      6 CH ." (B) SLOPE SENSITIVITY  = " SS @ SWAP / . CR
      6 CH ." (C) PEAK WIDTH         = " PW @ . CR
   CHOICE UNTIL SAVE-BUFFERS EMPTY-BUFFERS ;
-->


SCR # 51
( EXPERIMENT AND SAMPLE NAMES)

8 STRING DATE 40 STRING EXPERIMENT 14 STRING STD
14 STRING SMPLNAM 8 STRING SAMPLE# " SAMPLE #" SAMPLE# S!
VARIABLE SMPL# 1 SMPL# !
: SCRSETUP HOME 12 CV ;
: DATEENTRY SCRSETUP ." ENTER TODAY'S DATE (MM/DD/YY)  " INPUTS
      DATE S! ;
: EXPTENTRY SCRSETUP ." ENTER NAME OF EXPERIMENT" CR INPUTS
      EXPERIMENT S! ;
: LABELS DATEENTRY EXPTENTRY PRINTERON 14 EMIT 5 SPACES
      EXPERIMENT TYPE CR 14 EMIT 5 SPACES DATE  TYPE CR CR
      PRINTEROFF ;
: STDNAM? SCRSETUP " STD=" STD S! STD TYPE INPUTS STD S+ STD ;
-->


SCR # 52
( EXPERIMENT AND SAMPLE NAMES--CONT'D)

: MAKESMPLNAM SAMPLE# SMPLNAM S! SMPL# @ S->D STR$ SMPLNAM S+ ;
: SMPLNAM? SCRSETUP MAKESMPLNAM SMPLNAM TYPE CR CR
      ." PRESS SPACE BAR TO ACCEPT" CR
      ." PRESS ANY OTHER KEY TO CHANGE  "  KEY 32 = NOT
      IF SCRSETUP SAMPLE# TYPE INPUT DROP SMPL# ! MAKESMPLNAM
      THEN SMPLNAM ;
: INCSMPL# 1 SMPL# +! ;
-->
```

```
SCR # 53
( CONTINUOUS OPERATION FOR DATA ACQUISITION)

: SAMPLING START GR&TIMMODE 0 BEGIN TAKEDATA GRAPHIT INTEGRATE
            ?KEY UNTIL KBDRESET TX ;
: SAMPLE SAMPLING SMPLNAM? PRINTRESULTS INCSMPL# PRINTEROFF ;
: STANDARD SAMPLING STDNAM? PRINTRESULTS PRINTEROFF ;
: BASEADJUST START GR&TIMMODE 0 BEGIN TAKEDATA GRAPHIT DROP ?KEY
                  UNTIL KBDRESET TX SCRRESET ;
: GCDATA BEGIN SCRRESET
     ." 1=SMPL    2=STND    3=BSLN    4=VRBL    5=LBLS " KEY 49 -
     GODO SAMPLE STANDARD BASEADJUST PARAMETERMENU
        LABELS THEN 0
    UNTIL ;
: GASCHROMATOGRAPH EMPTY-BUFFERS 60 LOAD GCDATA ;


SCR # 60
( INTEGRATION PARAMETERS)

10 10 05
DATABUNCH ! PW ! DATABUNCH @ * SS !
```